7 Two-Dimensional Steady-State Heat Conduction. Analytical Solutions

In order to solve steady-state heat conduction problems, we have employed in this chapter a well-known separation of variables method, which is an analytical method. We have derived formulas for two-dimensional temperature distribution in fins of an infinite and finite length and in the radiant tubes of boilers. A computational program was developed for determining temperature and heat flux in finite-length-fins.

Exercise 7.1 Temperature Distribution in an Infinitely Long Fin with Constant Thickness

Determine temperature distribution in an infinitely long fin, shown in Fig. 7.1, by means of separation of variables method.

Fin base temperature is T_b, while the temperature of a fin-surrounding medium is T_{cz}. Heat transfer coefficient α on the fin surface is constant.



Fig. 7.1. A diagram of an infinitely long fin

Solution

Due to a symmetry of the temperature field, only an upper half of the fin will be examined here. Temperature distribution is described by the following differential equation

$$\frac{\partial^2 T}{\partial x^2}+\frac{\partial^2 T}{\partial y^2}=0 \tag{1}$$

and boundary conditions

$$T(0,y)=T_b\,, \tag{2}$$

$$T(\infty,y)=T_{cz}\,, \tag{3}$$

$$\frac{\partial T}{\partial y}(x,0)=0\,, \tag{4}$$

$$-\lambda\frac{\partial T}{\partial y}(x,w)=\alpha\left[T(x,w)-T_{cz}\right]. \tag{5}$$

Once dimensionless variables are introduced, such as

- temperature

$$\Theta=\frac{T-T_{cz}}{T_b-T_{cz}}\,, \tag{6}$$

- coordinates

$$X=\frac{x}{w},\qquad Y=\frac{y}{w}\,, \tag{7}$$

and Biot number

$$Bi=\frac{\alpha w}{\lambda}\,, \tag{8}$$

problem (1)–(5) can be written in the dimensionless form:

$$\frac{\partial^2 \Theta}{\partial X^2}+\frac{\partial^2 \Theta}{\partial Y^2}=0, \tag{9}$$

$$\Theta(0,Y)=1, \tag{10}$$

$$\Theta(\infty,Y)=0, \tag{11}$$

$$\frac{\partial \Theta}{\partial Y}(X,0)=0, \tag{12}$$

$$\frac{\partial \Theta}{\partial Y}(X,1)+Bi\cdot\Theta(X,1)=0. \tag{13}$$

According to the separation of variables method, the solution has the form

$$\Theta(X,Y)=U(X)\cdot V(Y). \tag{14}$$

By substituting (14) into (9), one obtains

$$U''V+UV''=0, \tag{15}$$

which results in

$$\frac{U''}{U}=-\frac{V''}{V}=\mu^2, \quad \text{gdzie} \quad U''=\frac{d^2U}{dX^2},\ V''=\frac{d^2V}{dY^2}. \tag{16}$$

From (16), one obtains two differential equations

$$\frac{d^2U}{dX^2}-\mu^2U=0, \tag{17}$$

$$\frac{d^2V}{dY^2}+\mu^2V=0. \tag{18}$$

Boundary conditions for (17) are obtained after substituting (14) into conditions (10) and (11)

$$UV\big|_{X=0}=1, \tag{19}$$

$$U(\infty)=0. \tag{20}$$

Boundary conditions for function $V(Y)$ are obtained by substituting solution (14) into boundary conditions (12) and (13)

$$\frac{dV}{dY}(0)=0, \tag{21}$$

$$\frac{dV}{dY}(1)+Bi\cdot V(1)=0. \tag{22}$$

A general solution for (18) is the function

$$V=A\cos\mu Y+B\sin\mu Y. \tag{23}$$

Accounting for boundary condition (21), yelds $B = 0$. Next, after substituting (23) into (22), characteristic equation is obtained

$$-\mu \cdot \sin \mu + Bi \cdot \cos \mu = 0 \ , \tag{24}$$

which can be written in the form

$$\mathrm{ctg}\mu = \frac{1}{Bi}\mu \, . \tag{25}$$

Equation (25) has an infinite number of roots $\mu_n > 0$, $n = 1, 2, ...,$ which are the characteristic values of the problem in question. It is evident, therefore, that an infinite number of solutions exists for the Sturm-Liouville problem (18), (21), (22)

$$V_n = A_n \cos \mu_n Y, \qquad n = 1, 2, ... \tag{26}$$

A general solution for (17) has the form

$$U = Ce^{\mu X} + De^{-\mu X} \, . \tag{27}$$

A great number of solutions U exist, which satisfy (17)

$$U_n = C_n e^{\mu_n X} + D_n e^{-\mu_n X} \, . \tag{28}$$

From the boundary condition (20), it is clear that $C_n = 0$. None of the solutions (14)

$$\Theta_n = U_n(X) V_n(Y) = A_n \cos \mu_n Y \cdot D_n e^{-\mu_n X} \tag{29}$$

satisfy boundary condition (19). Once notation $C_n = A_n D_n$ is introduced, the solution will be searched for in the form of a linear combination of function (29), in a way that will satisfy heterogeneous boundary condition (10)

$$\Theta = \sum_{n=1}^{\infty} \Theta_n = \sum_{n=1}^{\infty} C_n e^{-\mu_n X} \cos \mu_n Y. \tag{30}$$

By substituting (30) into (10), one obtains

$$\sum_{n=1}^{\infty} C_n \cos \mu_n Y = 1 \, . \tag{31}$$

After multiplying both sides of the equation by $\cos \mu_m Y$ and by integrating them from 0 to 1, one obtains

$$\sum_{n=1}^{\infty} \int_0^1 C_n \cos \mu_n Y \cdot \cos \mu_m Y dY = \int_0^1 \cos \mu_m Y dY \, . \tag{32}$$

Since a set of characteristic functions is a set of orthogonal functions, which satisfy

$$\int_0^1 \cos \mu_n Y \cdot \cos \mu_m Y dY = 0 \qquad \text{dla} \qquad m \neq n, \tag{33}$$

therefore, for $m = n$ from (32), one obtains

$$C_n = \frac{\int_0^1 \cos \mu_n Y dY}{\int_0^1 \cos^2 \mu_n Y dY}, \tag{34}$$

hence, after integration one obtains

$$C_n = \frac{2 \sin \mu_n}{\mu_n + \sin \mu_n \cos \mu_n}. \tag{35}$$

By substituting (35) into (30), a formula for temperature distribution has the form

$$\Theta(X,Y) = 2\sum_{n=1}^{\infty} \frac{\sin \mu_n}{\mu_n + \sin \mu_n \cos \mu_n} e^{-\mu_n X} \cos \mu_n Y.$$

Exercise 7.2 Temperature Distribution in a Straight Fin with Constant Thickness and Insulated Tip

Determine two-dimensional steady-state temperature distribution in a straight fin with thickness $2w$ and length l, made of a material with a constant thermal conductivity λ. The fin is secured to a surface with a constant temperature T_b. The fin tip is very well insulated. Lateral fin surfaces exchange heat with surroundings, at temperature T_{cz}, by convection when heat transfer coefficient α remains constant (Fig. 7.2).

Solution

Articles on the calculation of two-dimensional fin temperature fields are rather extensive in scope [2–9, 13, 14]. Fin temperature distribution is described by the heat conduction equation

$$\frac{\partial^2 T}{\partial x^2} + \frac{\partial^2 T}{\partial y^2} = 0, \tag{1}$$



Fig. 7.2. A fin diagram with an assumed coordinate system

when boundary conditions are

$$T(0,y)=T_b\,, \tag{2}$$

$$\frac{\partial T}{\partial x}(l,y)=0\,, \tag{3}$$

$$\frac{\partial T}{\partial y}(x,0)=0\,, \tag{4}$$

$$-\lambda\frac{\partial T}{\partial y}(x,w)=\alpha\left[T(x,w)-T_{cz}\right]. \tag{5}$$

After introducing dimensionless variables:

- temperature

$$\Theta=\frac{T-T_{cz}}{T_b-T_{cz}}\,, \tag{6}$$

- coordinates

$$X=\frac{x}{w},\qquad Y=\frac{y}{w}\,, \tag{7}$$

and the Biot number

$$Bi=\frac{\alpha w}{\lambda}\,, \tag{8}$$

problem (1)–(5) can be written in the dimensionless form:

$$\frac{\partial^2\Theta}{\partial X^2}+\frac{\partial^2\Theta}{\partial Y^2}=0\,, \tag{9}$$

$$\Theta(0,Y)=1, \tag{10}$$

$$\frac{\partial \Theta}{\partial X}(L,Y)=0, \tag{11}$$

$$\frac{\partial \Theta}{\partial Y}(X,0)=0, \tag{12}$$

$$\frac{\partial \Theta}{\partial Y}(X,1)+Bi\cdot\Theta(X,1)=0. \tag{13}$$

In accordance with the separation of variables method, the solution of problems (9)–(13) is searched for in the form

$$\Theta(X,Y)=U(X)\cdot V(Y). \tag{14}$$

Function $V(Y)$ has the same form as it does in Ex. 7.1. Function $U(X)$ is, much like in Ex. 7.1, determined from equation

$$\frac{\partial^2 U}{\partial X^2}-\mu^2 U=0, \tag{15}$$

when boundary conditions are

$$UV\big|_{X=0}=1, \tag{16}$$

$$\frac{\partial U}{\partial X}(L)=0, \text{ where } L=l/w. \tag{17}$$

The solution for (15), which satisfies condition (17), is the function

$$U=C\cosh\mu(L-X). \tag{18}$$

Since the characteristic equation (25) from Ex. 7.1 has an infinite number of positive elements μ_n, an infinite number of functions exist

$$U_n=C_n\cosh\mu_n(L-X), \qquad n=1,2,... \tag{19}$$

which satisfy (15) and boundary condition (17). The solution of problems (9)–(13) will be searched for in the form

$$\Theta(X,Y)=\sum_{n=1}^{\infty}U_nV_n=\sum_{n=1}^{\infty}C_n\cosh\mu_n(L-X)\cdot\cos\mu_nY. \tag{20}$$

Constant C_n in expression (20) is determined from the heterogeneous boundary condition (10), which yields the following result:

$$\sum_{n=1}^{\infty}C_n\cosh\mu_nL\cdot\cos\mu_nY=1. \tag{21}$$

After multiplying both sides of the equation by $\cos \mu_m Y$ and after integrating dY from 0 to 1, one gets

$$\sum_{n=1}^{\infty} \int_0^1 C_n \cosh \mu_n L \cdot \cos \mu_n Y \cdot \cos \mu_m Y = \int_0^1 \cos \mu_m Y dY . \tag{22}$$

From orthogonal condition of function $\cos \mu_n Y \cdot \cos \mu_m Y$ ((33), Ex. 7.1), one obtains

$$C_n \cosh \mu_n L = \frac{\int_0^1 \cos \mu_n Y dY}{\int_0^1 \cos^2 \mu_n Y dY} , \tag{23}$$

hence

$$C_n = \frac{2 \sin \mu_n}{\cosh \mu_n L \left(\mu_n + \sin \mu_n \cos \mu_n \right)} . \tag{24}$$

By substituting constant C_n formulated in (24) into expression (20), temperature distribution is formulated as

$$\begin{aligned} \Theta(X,Y) = 2 \sum_{n=1}^{\infty} \frac{\sin \mu_n}{\cosh \mu_n L \left(\mu_n + \sin \mu_n \cos \mu_n \right)} \times \\ \times \cosh \mu_n \left(L - X \right) \cdot \cos \mu_n Y, \end{aligned} \tag{25}$$

where μ_n are positive elements of the characteristic transcendental equation

$$\operatorname{ctg} \mu = \frac{1}{Bi} \mu . \tag{26}$$

Exercise 7.3 Calculating Temperature Distribution and Heat Flux in a Straight Fin with Constant Thickness and Insulated Tip

Calculate temperature distribution in a fin on the basis of a formula derived in Ex. 7.2. Calculate fin temperature in points shown in Fig. 7.3. Also calculate heat flux at the fin base in points (0,0) and (0,w). Determine a formula for averaged temperature and heat flux across the fin thickness. Compare mean temperature values across the entire fin length and heat flux at the fin base with one-dimensional solution. Assume the following values for the calculation: w = 0.003m, l = 0.024 m, α = 100 W/(m^2·K), T_b = 95°C, T_{cz} = 20°C, Δx = 0.003 m, λ = 50 W/(m·K).



Fig. 7.3. A fin diagram with marked nodes, in which temperature is calculated

Solution

On the basis of (25) from Ex. 7.2, temperature distribution $T\,(X,Y)$ will be calculated from formula

$$T = T_{cz} + \left(T_b - T_{cz}\right)\Theta, \tag{1}$$

where

$$\Theta\left(X,Y\right) = 2\sum_{n=1}^{\infty} \frac{\sin\mu_n}{\left(\mu_n + \sin\mu_n \cdot \cos\mu_n\right)} \frac{\cosh\mu_n\left(L-X\right)}{\cosh\mu_n L} \cos\mu_n Y, \tag{2}$$

where: $X = x/w$, $Y = y/w$, $Bi = \alpha w/\lambda$.

Elements of the characteristic equation

$$\text{ctg}\mu = \frac{1}{Bi}\mu \tag{3}$$

will be determined by means of the interval halving method; one should note, however, that the infinite element μ_n lies in an interval between $\mu_{n,\min} = (n-1)\pi$ and $\mu_{n,\max} = (n-1/2)\pi$. Values $\mu_{n,\max}$ are characteristic values of the Strum-Liouville problem in an instance when $Bi \to \infty$, i.e. when constant temperature is assigned on the fin surface.

Mean temperature across the fin thickness is determined from formula

$$\bar{T}(x) = \frac{1}{w}\int_0^w \left[T_{cz} + \left(T_b - T_{cz}\right)\theta\right]dy, \tag{4}$$

from where, one obtains

$$\overline{T} = T_{cz} + 2\left(T_b - T_{cz}\right)\sum_{n=1}^{\infty} \frac{\sin\mu_n}{\mu_n\left(\mu_n + \sin\mu_n \cdot \cos\mu_n\right)} \frac{\cosh\mu_n\left(L - X\right)}{\cosh\mu_n L}\sin\mu_n . \quad (5)$$

Heat flux in the direction of x axis comes to

$$\dot{q}_x = -\lambda\frac{\partial T}{\partial x} = \frac{2\lambda\left(T_b - T_{cz}\right)}{w}\sum_{n=1}^{\infty} \frac{\mu_n \sin\mu_n}{\left(\mu_n + \sin\mu_n \cdot \cos\mu_n\right)} \frac{\sinh\mu_n\left(L - X\right)}{\cosh\mu_n L} \times \quad (6)$$
$$\times \cos\mu_n Y .$$

Mean heat flux value across the fin thickness is the function of x coordinate

$$\overline{\dot{q}}_x = \frac{1}{w}\int_0^w \dot{q}_x\left(x,y\right)dy = \frac{2\lambda\left(T_b - T_{cz}\right)}{w}\sum_{n=1}^{\infty} \frac{\sin\mu_n}{\left(\mu_n + \sin\mu_n \cdot \cos\mu_n\right)} \times$$
$$\times \frac{\sinh\mu_n\left(L - X\right)}{\cosh\mu_n L}\sin\mu_n . \quad (7)$$

Fin temperature distribution T_{1d}, determined under the assumption that temperature decrease within the fin thickness is negligibly small, is expressed by function (Ex. 6.15)

$$T_{1d}\left(x\right) = T_{cz} + \left(T_b - T_{cz}\right)\frac{\cosh m\left(l - x\right)}{\cosh ml} , \quad (8)$$

where $m = \sqrt{\alpha/\lambda w}$.

In the given case

$$m = \sqrt{\frac{50}{100 \cdot 0.003}} = 12.90994 \text{ 1/m} .$$

In the case of the one-dimensional solution, heat flux is formulated as

$$\dot{q}_{1d}\left(x\right) = -\lambda\frac{dT_{1d}}{dx} = \left(T_b - T_{cz}\right)\lambda m\frac{\sinh m\left(l - x\right)}{\cosh ml} . \quad (9)$$

Allowing that $Bi = \alpha w/\lambda = 100 \cdot 0.003/50 = 0.006$, the first ten elements of the characteristic equation (3) were determined:

μ_1	μ_2	μ_3	μ_4	μ_5	μ_6	μ_7	μ_8	μ_9	μ_{10}
0.0774	3.1435	6.2841	9.4254	12.5668	15.7083	18.8499	21.9914	25.1330	28.2745

Next, the elements were applied to (2). Temperature $T(x, y)$ in nodes shown in Fig. 7.3 was calculated by means of the FORTRAN program, which comes with this exercise. Proper values μ_n were calculated by means of the sub-program presented in paper [1]. Mean temperature distribution $\overline{T}(x)$ and temperature $T_{1d}(x)$ were also calculated.

Heat flux was calculated at two points: B_1 and B_2 (Fig. 7.3). Mean heat flux $\overline{\dot{q}_x}$ at the fin base ($X = 0$) was calculated on the basis of (7). For comparison purposes $\dot{q}_{1d}$ was also calculated for $x = 0$ by means of (9). Temperature calculation results are shown in Table 7.1.

Table 7.1. Calculation results

x [m]	Node no.	Temperature [°C]	Node no.	Temperature [°C]	Node no.	$\overline{T}(x)$ [°C]	$T_{1d}(x)$ [°C]
0	B_1	95.00	B_2	94.99	B_1, B_2	95.00	95.00
0.003	1	92.09	2	91.88	1, 2	92.02	92.02
0.006	3	89.55	4	89.34	3, 4	89.48	89.47
0.009	5	87.42	6	87.22	5, 6	87.35	87.34
0.012	7	85.70	8	85.50	7, 8	85.63	85.62
0.015	9	84.37	10	84.18	9, 10	84.30	84.29
0.018	11	83.42	12	83.23	11, 12	83.36	83.34
0.021	13	82.86	14	82.67	13, 14	82.80	82.78
0.024	15	82.67	16	82.48	15, 16	82.61	82.59

Calculated heat flux measures:

- at point B_1

  $$\dot{q}_x(0,0) = 49764 \text{ W/m}^2,$$

- at point B_2

  $$\dot{q}_x(0,w) = 66646 \text{ W/m}^2,$$

- mean heat flux at the fin base $\overline{\dot{q}_x}$

  $$\overline{\dot{q}_x}(0) = 53253 \text{ W/m}^2,$$

- heat flux $\dot{q}_{1d}$ at the fin base

  $$\dot{q}_{1d}(0) = 53341 \text{ W/m}^2.$$

From the analysis of the obtained results, it is evident that there is a small temperature decrease across the fin thickness. Also, heat flux $\dot{q}_x$ varies in points B_1 and B_2. A good accuracy of results is evident in $\overline{T}(x)$ and $T_{1d}(x)$,

$\overline{\dot{q}_x}$ (0) and $\dot{q}_{1d}$ (0), i.e. between the mean values across the fin thickness obtained under the assumption that fin temperature field is two-dimensional and between values determined under the assumption that temperature decrease across the fin thickness is negligibly small, i.e. temperature and heat flux are only the function of x coordinate.

Program for Calculating Two-Dimensional Fin Temperature Field

```
      program fin
      dimension eigen(50)
      open(unit=1,file='fin.in')
      open(unit=2,file='fin.out')
      read(1,*)ne,bi
      read(1,*)t_cz,t_b,dlug,w,s_lam,s_alfa
      write(2,'(a)')
     &"CALCULATING TWO-DIMENSIONAL FIN TEMPERATURE FIELD"
      write(2,'(/a)') "DATA ENTERED"
      write(2,'(a,i10)')   "ne     =",ne
      write(2,'(a,e10.5)') "Biot number=",bi
      write(2,'(a,e10.5,a)') "t_cz   =",t_cz," [C]"
      write(2,'(a,e10.5,a)') "t_b    =",t_b," [C]"
      write(2,'(a,e10.5,a)') "dlug   =",dlug," [m]"
      write(2,'(a,e10.5,a)') "w      =",w," [m]"
      write(2,'(a,e10.5,a)') "lambda =",s_lam," [W/mK]"
      write(2,'(a,e10.5,a)') "alfa   =",s_alfa," [W/m2K]"
      write(2,'(/a,i3,a)') "CALCULATION OF FIRST",ne,
     &" EQUATION ELEMENTS X*TAN(X)=BI"
      call equation_elements (bi,ne,eigen)
      write(2,'(/a)')"CALCULATED EQUATION ELEMENTS"
      write(2,'(a)')"Lp         mi"
      do i=1,ne
        write(2,'(i2,5x,e11.6)') i,eigen(i)
      enddo
      write(2,'(/a)')"CALCULATED TEMPERATURE [C]"
      write(2,'(a)')"  x[m] T(x,B1)  T(x,B2) T_sr(x) T_1d(x)"
      x=0.
      do i=1,10
       write(2,'(f5.3,4(3x,e10.5))')x,
     & temperature(x,0.,t_cz,t_b,dlug,w,ne,eigen),
     & temperature(x,w,t_cz,t_b,dlug,w,ne,eigen),
     & temperature_sr(x,t_cz,t_b,dlug,w,ne,eigen),
     & temperature_1d(x,t_cz,t_b,dlug,w,s_lam,s_alfa)
       x=x+dlug/float(8)
      enddo
      write(2,'(/a)')"CALCULATED HEAT FLUX [W/m2]"
      write(2,'(a,e10.5)') "q_x(0,0)=",
     &value_q(0.,0.,t_cz,t_b,dlug,w,ne,eigen,s_lam)
```

```
 write(2,'(a,e10.5)') "q_x(0,w)=",
&value_q(0.,w,t_cz,t_b,dlug,w,ne,eigen,s_lam)
 write(2,'(a,e10.5)') "q_x_sr(0)=",
&value_q_sr(0.,t_cz,t_b,dlug,w,ne,eigen,s_lam)
 write(2,'(a,e10.5)') "q_x_1d(0)=",
&value_q_1d(0.,t_cz,t_b,dlug,w,s_lam,s_alfa)
 end  program fin

 function value_q_1d(x,t_cz,t_b,dlug,w,s_lam,s_alfa)
 s_m=sqrt(s_alfa/s_lam/w)
 value_q_1d=(t_b-t_cz)*s_lam*s_m*sinh(s_m*
&(dlug-x))/cosh(s_m*dlug)
 end function

 function value_q(x,y,t_cz,t_b,dlug,w,ne,eigen,s_lam)
 dimension eigen(*)
 teta=0.
 x_b=x/w
 y_b=y/w
 dlug_b=dlug/w
 do i=1,ne
 s=eigen(i)
 teta=teta+s*sin(s)*sinh(s*(dlug_b-x b))
&*cos(s*y_b)/(s+sin(s)*cos(s))/cosh(s*dlug_b)
 enddo
 value_q=2.*(t_b-t_cz)*s_lam*teta/w
 end function
 function value_q_sr(x,t_cz,t_b,dlug,w,ne,eigen,s_lam)
 dimension eigen(*)
 teta=0.
 x_b=x/w
 dlug_b=dlug/w
 do i=1,ne
 s=eigen(i)
 teta=teta+sin(s)*sinh(s*(dlug_b-x_b))*sin(s)/
&(s+sin(s)*cos(s))/cosh(s*dlug_b)
 enddo
 value_q_sr=2.*(t_b-t_cz)*s_lam*teta/w
 end function

 function temperature_1d(x,t_cz,t_b,dlug,w,s_lam,s_alfa)
 s_m=sqrt(s_alfa/s_lam/w)
 temperature_1d=t_cz+(t_b-t_cz)*cosh(s_m*(dlug-x))
&/cosh(s_m*dlug)
 end function

 function temperature_sr(x,t_cz,t_b,dlug,w,ne,eigen)
```

```
      dimension eigen(*)
      teta=0.
      x_b=x/w
      dlug_b=dlug/w
      do i=1,ne
      s=eigen(i)
      teta=teta+sin(s)*cosh(s*(dlug_b-x_b))*sin(s)/
     &s/(s+sin(s)*cos(s))/cosh(s*dlug_b)
      enddo
      temperature_sr=t_cz+(t_b-t_cz)*2.*teta
      end function

      function temperature(x,y,t_cz,t_b,dlug,w,ne,eigen)
      dimension eigen(*)
      teta=0.
      x_b=x/w
      y_b=y/w
      dlug_b=dlug/w
      do i=1,ne
         s=eigen(i)
      teta=teta+sin(s)*cosh(s*(dlug_b-x_b))*cos(s*y_b)/
     &(s+sin(s)*cos(s))/cosh(s*dlug_b)
      enddo
      temperature=t_cz+(t_b-t_cz)*2.*teta
      end function

c     procedure calculates elements of characteristic eq.
c     x*tan(x)=bi where bi is Biot number, ne calculated
c     element quantity, eigen vector with recorded calculated
c     elements
      subroutine equation_elements(bi,ne,eigen)
      dimension eigen(*)
      pi=3.141592654
      do i=1,ne
         xi=(float(i)-1.)*pi
         xf=pi*(float(i)-.5)
         do while (abs(xf-xi).ge.5.E-06)
            xm=(xi+xf)/2.
            y=xm*sin(xm)/cos(xm)-bi
            if (y.lt.0.) then
               xi=xm
            else
               xf=xm
            endif
         enddo
        eigen(i)=xm
      enddo
      return
```

```
      end
data(fin.in)
10 0.006
20. 95. 0.024 0.003  50. 100.
results(fin.out)
CALCULATING TWO-DIMENSIONAL FIN TEMPERATURE FIELD

DATA ENTERED
ne      =         10
Biot number=.60000E-02
t_cz    =.20000E+02 [C]
t_b     =.95000E+02 [C]
dlug    =.24000E-01 [m]
w       =.30000E-02 [m]
lambda  =.50000E+02 [W/mK]
alfa    =.10000E+03 [W/m2K]

CALCULATION OF FIRST 10 EQUATION ELEMENTS X*TAN(X)=BI

CALCULATED EQUATION ELEMENTS
Lp          mi
 1      .773851E-01
 2      .314350E+01
 3      .628414E+01
 4      .942542E+01
 5      .125668E+02
 6      .157083E+02
 7      .188499E+02
 8      .219914E+02
 9      .251330E+02
10      .282745E+02

CALCULATED TEMPERATURE [C]
  x[m]   T(x,B1)      T(x,B2)     T_sr(x)     T_1d(x)
 .000   .95000E+02  .94990E+02 .95000E+02 .95000E+02
 .003   .92095E+02  .91887E+02 .92026E+02 .92021E+02
 .006   .89554E+02  .89346E+02 .89485E+02 .89475E+02
 .009   .87426E+02  .87224E+02 .87359E+02 .87346E+02
 .012   .85702E+02  .85506E+02 .85637E+02 .85621E+02
 .015   .84372E+02  .84180E+02 .84308E+02 .84290E+02
 .018   .83428E+02  .83238E+02 .83365E+02 .83345E+02
 .021   .82863E+02  .82675E+02 .82801E+02 .82781E+02
 .024   .82676E+02  .82488E+02 .82613E+02 .82593E+02
 .027   .82863E+02  .82675E+02 .82801E+02 .82781E+02

CALCULATED HEAT FLUX [W/m2]
q_x(0,0)=.49764E+05
q_x(0,w)=.66646E+05
q_x_sr(0)=.53253E+05
q_x_1d(0)=.53341E+05
```

Exercise 7.4 Temperature Distribution in a Radiant Tube of a Boiler

Determine formula for temperature distribution in the boiler's radiant tube (Fig. 7.4) by means of the separation of variables method. Assuming that heat flux $\dot{q}_m$ (thermal load of the water-wall) transferred by the water-wall is known (calculated with reference to a wall regarded as a plane), as well as the temperature of a medium that flows inside the tube T_{cz} and heat transfer coefficient α on an inner surface of the tube, determine temperature field in the function of coordinates r and φ. Also calculate the inner and outer surface tube temperature for angle $\varphi = 0$ and $\varphi = \pi$ rad; use the following values for the calculation:

- outer surface tube radius $r_z = 0.019$ m,
- inner surface tube radius $r_w = 0.015$ m,
- scale of radiant tube spacing $s = 0.042$ m,
- thermal load of the water-wall $\dot{q}_m = 300000$ W/m^2;
- heat transfer coefficient on the inner surface of the tube $\alpha = 15\,000$ W/(m^2·K),
- temperature of a medium $T_{cz} = 330$°C,
- heat conduction coefficient of the steel which the tube is made of $\lambda = 45$ W/(m·K).



Fig. 7.4. A diagram of a radiant tube spacing in a combustion chamber

Heat flux on the outer surface of the tube is expressed by function [12]

$$\dot{q}(\varphi)=\dot{q}_m\left(0.3649+0.4777\cos\varphi+0.1574\cos 2\varphi\right) . \tag{1}$$

Solution

Tube temperature distribution is expressed by heat conduction equation

$$\frac{1}{r}\frac{\partial}{\partial r}\left(r\frac{\partial T}{\partial r}\right)+\frac{1}{r^2}\frac{\partial^2 T}{\partial\varphi^2}=0 \tag{2}$$

and boundary conditions

$$\lambda\frac{\partial T}{\partial r}\bigg|_{r=r_z}=\dot{q}_0+\sum_{n=1}^{\infty}\dot{q}_n\cos(n\varphi) , \tag{3}$$

$$\lambda\frac{\partial T}{\partial r}\bigg|_{r=r_w}=\alpha T\big|_{r=r_w} , \tag{4}$$

where T is the temperature excess of the tube ζ above the temperature of the medium ζ_{cz}, i.e. $T=\zeta-\zeta_{cz}$.

In conformity with the separation of variables method, the solution is searched for in the form

$$T(r,\varphi)=U(r)\cdot V(\varphi) . \tag{5}$$

By substituting (5), one obtains equation

$$r^2U''V+rU'V+UV''=0 . \tag{6}$$

After a division of (6) by UV and the separation of variables, one obtains

$$\frac{r^2U''+rU'}{U}=-\frac{V''}{V} . \tag{7}$$

Since r and φ are independent variables, equality (7) occurs only when its both sides are equal to the same constant. If the constant were negative, the solution $V(\varphi)$ would then contain exponential functions, which would unable one to satisfy periodic boundary condition (3) written in the Fourier series form. Separated constant, therefore, must be either a positive integral number or zero. If one assumes that both sides of (7) are equal to n^2, one obtains

$$r^2U'' + rU' - n^2U = 0, \tag{8}$$

$$V'' + n^2V = 0, \qquad n = 0, 1, \ldots \tag{9}$$

In the case of a circular-symmetrical load only $\dot{q}_0 \neq 0$, whereas $\dot{q}_1 = \dot{q}_2 = \cdots = 0$. For $n = 0$, the solution of (8) and (9) has the form

$$U(r) = A_0' + B_0' \ln r \tag{10}$$

and

$$V(\varphi) = C_0' + D_0'\varphi . \tag{11}$$

Due to the circular-symmetrical load $D_0' = 0$, the product $U(r)V(\varphi)$ can be written in the form

$$U(r)V(\varphi) = A_0 + B_0 \ln r, \qquad n = 0 , \tag{12}$$

where, $A_0 = A_0'C'$, $B_0 = B_0'C_0'$.

For $n \geq 1$, the solution of (8)–(9) has the forms

$$U(r) = A_n' r^n + B_n' r^{-n} , \tag{13}$$

$$V(\varphi) = C_n' \cos n\varphi + D_n' \sin n\varphi . \tag{14}$$

Due to the symmetry of tube heating (condition (3)) with respect to the plane, which is perpendicular to the water-wall and which crosses the tube axis, constant $D_n' = 0$. Product $U(r)V(\varphi)$ can be written in the form

$$U(r)V(\varphi) = \left(C_n r^n + D_n r^{-n}\right)\cos n\varphi, \qquad n \geq 1, \tag{15}$$

where, $C_n = A_n' C_n'$ and $D_n = B_n' C_n'$.

Expression (5), which describes the distribution of excess temperature in the tube, has the form

$$T(r,\varphi) = A_0 + B_0 \ln r + \sum_{n=1}^{\infty}\left(C_n r^n + D_n r^{-n}\right)\cos n\varphi . \tag{16}$$

After substituting (16) into boundary conditions (3) and (4), one can determine constants, which can be written after transformation in the following form:

$$A_0 = \frac{\dot{q}_0 r_z}{\lambda}\left(\frac{1}{Bi} - \ln r_w\right),$$

$$B_0 = \frac{\dot{q}_0 r_z}{\lambda},$$

$$C_n = \frac{\dot{q}_n r_z}{\lambda} \frac{\frac{1}{n} u^n (Bi + n) \frac{1}{r_w^n}}{Bi(u^{2n}+1) + n(u^{2n}-1)},$$

$$D_n = -\frac{\dot{q}_n r_z}{\lambda} \frac{\frac{1}{n} u^n (Bi - n) r_w^n}{Bi(u^{2n}+1) + n(u^{2n}-1)},$$

where, $u = r_z/r_w$, $Bi = \alpha\, r_w/\lambda$.

Since in this exercise, the heat flux on the outer surface of the pipe is defined by (1), then

$$\dot{q}_0 = 0.3649\dot{q}_m,$$

$$\dot{q}_1 = 0.4777\dot{q}_m,$$

$$\dot{q}_2 = 0.1574\dot{q}_m,$$

and

$$\dot{q}_3 = \dot{q}_4 = ... = 0,$$

It is easy to calculate tube temperature, when only 2 terms are accounted for in the series (16). Once the following is calculated

$$u = \frac{r_z}{r_w} = \frac{0.019}{0.015} = 1.2667,$$

$$Bi = \frac{\alpha r_w}{\lambda} = \frac{15000 \cdot 0.015}{45} = 5.0$$

and substituted into solution (16), one obtains

$$T(r_z, 0) = 52.28°\text{C},$$

$$T(r_w, 0) = 23.53°\text{C},$$

$$T(r_z, \pi) = 2.31°\text{C},$$

$$T(r_w, \pi) = 1.04°\text{C}.$$

These are temperatures above the medium's temperature. Corresponding pipe temperatures are:

$$\zeta(r_z, 0) = T_{cz} + T(r_z, 0) = 330 + 52.28 = 382.28°\text{C},$$

$$\zeta(r_w, 0) = T_{cz} + T(r_w, 0) = 330 + 23.53 = 353.53°\text{C},$$

$$\zeta(r_z,\pi)=T_{cz}+T(r_z,\pi)=330+2.31=332.31°\mathrm{C},$$
$$\zeta(r_w,\pi)=T_{cz}+T(r_w,\pi)=330+1.04=331.04°\mathrm{C}.$$

It is evident that temperature $\zeta(r_z,0)$ is the maximum temperature across the whole cross-section of the tube. Provided that this temperature is known, one can correctly chose the right type of steel for the radiant tube of a boiler.

Literature

1. Becker M (1986) Heat Transfer. A Modern Approach. Plenum Press, New York
2. Gdula SJ et. al. (1984) Heat conduction. Group work. PWN, Warsaw: 134–138
3. Irey RK (1968) Errors in the one-dimensional fin solution. Transactions of the ASME, J. Heat Transfer 90: 175–178
4. Lau W, Tan CW (1973) Errors in one-dimensional heat transfer analyses in straight and annular fins. Transactions of the ASME, J. Heat Transfer 95: 549–551
5. Look DC (1995) Fin on pipe (insulated tip): minimum conditions for fin to be beneficial. Heat Transfer Engineering 16, No. 3: 65–75
6. Look DCJr, Kang HS (1991) Effects of variation in root temperature on heat lost from a thermally non-symmetric fin. Int. J. Heat Mass Transfer 34, No. 4/5: 1059–1065
7. Look DCJr (1999) Fin (on pipe) effectiveness: one dimensional and two dimensional. Transactions of the ASME, J. Heat Transfer 121: 227–230
8. Look DCJr (1988) Two-dimensional fin performance: Bi (top surface) > Bi (bottom surface). Transactions of the ASME, J. Heat Transfer 110: 780–782
9. Ma SW, Behbahani AI, Tsuei YG (1991) Two-dimensional rectangular fin with variable heat transfer coefficient. Int. J. Heat and Mass Transfer 34, No 1: 79–85
10. Młynarski F, Taler J (1979) Analytische Untersuchung der Temperatur – und Spannungsverteilung in strahlungsbeheizten Kesselrohren unter der Berücksichtigung der wasserseitigen Ablagerungen. VGB Kraftwerkstechnik 5: 440–447
11. Taler J, Kulesza L (1982) Calculating temperature field in finned, bimetallic and homogenous screen pipes with a layer of residue on an inner surface. Archives of Power Engineering 1: 3–23
12. Taler J (1980) Weighed residues method and its application for temperature field calculation in boilers' elements. Monograph 14, Pub. Krakow University of Technology, Kraków
13. (1999) The Heat Transfer Problem Solver, Problem 3–5. Research and Education Association. Piscataway, New Jersey USA: 142–145
14. Unal HC (1988) The effect of the boundary condition at a fin tip on the performance of the fin with and without internal heat generation. Int. J. Heat Mass Transfer 31, No. 7: 1483–1496
15. Kraus AD, Aziz A, Welty J (2001) Extended Surface Heat Transfer. Wiley & Sons, New York